

Mãos na massa: Rotinas internas

Chegou a hora de você pôr em prática o que foi visto na aula. Para isso, execute os passos listados abaixo.

- 1) Adicione o seguinte código, das funções `OBTER_CATEGORIA_CLIENTE` e `VERIFICA_SEGMENTO_MERCADO` e a *procedure* `FORMAT_CNPJ`, no **início do corpo** do pacote `CLIENTE_PAC`:

```
FUNCTION OBTER_CATEGORIA_CLIENTE (
    p_faturamento_previsto IN cliente.faturamento_previsto%TYPE
) RETURN cliente.categoria%TYPE
IS
BEGIN
    IF
        p_faturamento_previsto < 10000
    THEN
        RETURN 'PEQUENO';
    ELSIF p_faturamento_previsto < 50000 THEN
        RETURN 'MEDIO';
    ELSIF p_faturamento_previsto < 100000 THEN
        RETURN 'MEDIO GRANDE';
    ELSE
        RETURN 'GRANDE';
    END IF;
END;

PROCEDURE FORMAT_CNPJ (
    p_cnpj IN OUT VARCHAR2
)
IS
BEGIN
    p_cnpj := substr(p_cnpj,1,2)
    || '/'
    || substr(p_cnpj,3);
END;

FUNCTION VERIFICA_SEGMENTO_MERCADO
    (p_id IN segmercado.id%type)
    RETURN boolean
IS
    v_dummy number(1);
BEGIN
    SELECT 1 INTO v_dummy
    FROM segmercado
    WHERE id = p_id;
    RETURN true;
EXCEPTION
    WHEN no_data_found THEN
        RETURN false;
END;
```

- 2) Em `user_dev`, para descobrir quais rotinas utilizam a *procedure* `INCLUIR_CLIENTE` de fora do pacote, veja as suas dependências:

```
EXEC DEPTREE_FILL('procedure','USER_DEV','incluir_cliente')
```

3) E analise os dados:

```
SELECT NESTED_LEVEL,SCHEMA,TYPE,NAME FROM DEPTREE  
ORDER BY SEQ#
```

4) Analisando os dados, você consegue ver que a *procedure* `APL_INCLUIR_CLIENTE`, do usuário `user_app`, faz referência à *procedure* `INCLUIR_CLIENTE` de fora do pacote. Então, em `user_app`, modifique a *procedure* `APL_INCLUIR_CLIENTE` para fazer referência à *procedure* `INCLUIR_CLIENTE` de **dentro** do pacote:

```
create or replace PROCEDURE APL_INCLUIR_CLIENTE  
  (p_id in cliente.id%type,  
   p_razao_social in cliente.razao_social%type,  
   p_CNPJ in cliente.CNPJ%type ,  
   p_segmercado_id in cliente.segmercado_id%type,  
   p_faturamento_previsto in cliente.faturamento_previsto%type)  
IS  
BEGIN  
  cliente_pac.incluir_cliente(p_id ,p_razao_social,p_cnpj,p_segmercado_id,p_faturamento_previsto);  
END;
```

5) Volte a `user_dev` e apague as rotinas que não são mais necessárias, que estão fora do pacote `CLIENTE_PAC` :

```
DROP PROCEDURE INCLUIR_CLIENTE;  
DROP PROCEDURE EXCLUIR_CLIENTE;  
DROP PROCEDURE ATUALIZAR_CLI_SEG_MERCADO;  
DROP PROCEDURE ATUALIZAR_FATURAMENTO_PREVISTO;  
DROP PROCEDURE FORMAT_CNPJ;  
DROP FUNCTION OBTER_CATEGORIA_CLIENTE;
```